

Nieuw in Uniface: DAC8-print, een snelle D/A convertor in enkele en dubbele uitvoering.

Hans Zeedijk

In PTC-Print van januari werd de ADC8 Uniface print bij U geïntroduceerd, maar ook nu is er weer nieuws: een uitbreiding van Uniface met een digitaal-analoog omzetter, dus in feite het complement van de ADC8-print. Wat doet dat ding: de computer stuurt een signaal tussen 0 en 255 naar de print en deze levert op zijn uitgang een spanning af tussen 0 en 2,55 Volt en wel voor iedere eenheid van het uitgestuurde getal 10 mV.

Met deze nieuwe print wordt het Uniface-systeem afgerond in die zin, dat nu alle vormen van communicatie met de buitenwereld mogelijk zijn: digitaal en analoog in en uit.

Net als de ADC8 heeft de DAC8 voldoende aan een voedingsspanning van 5 Volt, die door de computer geleverd wordt via de aangesloten bandkabel. De DAC8 is razendsnel met een conversietijd binnen 1,5 microseconde, zodat het gebruiksprogramma vrijwel altijd snelheidslimiterend zal zijn. Er zijn twee uitvoeringen van de DAC8 verkrijgbaar: een enkele uitvoering met één uitgang en een dubbele uitvoering met twee uitgangen. De print is overigens voor beide gelijk. De dubbele uitvoering kan van belang zijn, omdat analoge signalen niet zelden paarsgewijs gebruikt worden. Denk aan dingen als X en Y-as, of stereo.

De print

De plattegrond van de print toont het gebruikelijke beeld van het adresgedeelte (van onder tot en met de dipswitch) en het functionele gedeelte. Het adresdeel is wat uitvoeriger dan normaal, omdat bij de DAC8 twee bytes uitgestuurd moeten worden: 1 byte voor de data en 1 byte voor de aansturing van de conversie. Van die laatste byte worden maar 2 bits per DA-chip gebruikt voor resp. chip select en chip enable.

Addressering vindt weer plaats in 8 bits (dus 256 adressen) waarvan er 4 variabel zijn door instelling van de dipswitch. De print gebruikt wel twee adressen en de te kiezen adressen zijn: 64/65, 66/67, 68/69 en 70/71. In de programmering moet weer geïnverteerd worden (dus gebruiken resp. de waarden 191/190, 189/188, 187/186 en 185/184). De data komen in het hoogste adres, maar zie hiervoor ook het programmeer-voorbeeld.

De fabrikant van de conversie-chip (Analog Devices) geeft als nauwkeurigheid van de conversie 1/2 minst significant bit op, dus 5 mV op een schaal van 0 tot 2,55 Volt. Deze afwijking kan ook optreden in de lage uitgangsspanningen. Een zero offset van 5 mV is mogelijk. De enige DAC8 die ik tot nu getest heb bleef op beide uitgangen ruim binnen de specificaties van de fabrikant.

Voor wie is de print bestemd ?

De DAC8 is een mooi hulpmiddel voor diegenen, die eenvoudig instelbare en redelijk nauwkeurige ijkspanningen nodig heeft. Tussen 1 Volt en 2,55 Volt is de nauwkeurigheid van de afgegeven spanningen ruim binnen 1%. Door de uitgangsspanning volgens een bepaald patroon te variëren kan een spanningsfunctie naar eigen keuze gemaakt worden, waarvan in de volgende paragraaf een voorbeeld. En uiteraard kan de DAC8 gebruikt worden voor analoge uitsturing van buitenwereld-processen. Een aardige toepassing kan zijn het opwekken van geluidstrillingen en deze kunnen een exakte frekwentie hebben, zodat ze gebruikt kunnen worden voor het afstemmen van muziekinstrumenten. In combinatie met de ADC8 print en de DAC8 print kan geluidssignaal, b.v. afkomstig van een microfoon, gedigita-

liseerd worden, opgeslagen in het computergeheugen, en later weer afgespeeld. Hiermee heb ik al geëxperimenteerd bij een sample en afspeelfrekwentie van 8 kHz en dat levert alszins redelijke geluidskwaliteit op.

Programmering van de DAC8.

De print vereist twee adressen, b.v. 64 en 65 voor resp. de aansturing en de data. De aansturing geschiedt in de eerste twee bitparen. Wanneer deze op nul worden gebracht komt de byte die via de databus naar de print is weggeschreven op de uitgang van de DAC8 als een spanning. Stel we willen een spanning van 1 Volt op de uitgang realiseren via programmering in BASIC op een MSX computer. Dan zijn de volgende handelingen nodig:

regel 1/cw vrh/: OUT 48,100:OUT 49,191

/cw/De waarde 100 staat nu in het data-register van de DAC8.

regel 2/cw/: OUT 49,190:OUT 48,0

/cw/De uitgang van de DAC8 geeft nu 1 Volt af. Wanneer alleen uitgang 1 nodig is van de dubbele uitvoering gebruikt men de waarde 12 i.p.v. 0; en voor alleen uitgang 2 de waarde 3 i.p.v. 0. Voor andere computers moet de output-poort anders gekozen worden: 96 en 97 i.p.v. 48 en 49 voor de P2000; en voor de PC resp. 784 en 785.

/cw/Veranderingen op de databus uitgestuurd zoals in regel 1 worden aan de DAC8 uitgang doorgegeven zolang de waarde 0 gehandhaafd blijft. Als dit niet mag moeten de bits 0 t/m 3 weer hoog gemaakt worden volgens:

/cw/regel 3: OUT 49,190:OUT 48,15

/cw/De spanning op de uitgang van de DAC8 blijft nu behouden en wijzigingen op de databus hebben daar geen invloed meer op.

Het programma FUNGEN.BAS.


Als voorbeeld van een toepassing van de DAC8 print wordt FUNGEN.BAS gegeven, dat een spanningsfunctie op de DAC8 uitgang naar keuze genereert. De te kiezen functies zijn: een sinus, een blok, twee zaagtanden en een driehoek. De amplitude van de spanningsfunctie kan gevarieerd worden en tevens de frekwentie.

Het programma lijkt lang, maar het grootste gedeelte bestaat uit keuzemenu's (regel 120 t/m 200 voor de functie

vorm, 210 t/m 270 voor de amplitude en 300 t/m 340 voor de frekwentie) en het genereren van de gekozen functie, die opgeslagen wordt in de tabel FUN(100)(regel 400 t/m 440 voor de sinus, 500 t/m 530 voor de blok, 600 t/m 620 voor de zaagtand oplopend, 700 t/m 720 voor de zaagtand aflopend, en 800 t/m 830 voor de driehoek). Deze delen van het programma spreken voor zich.

De DAC8 wordt geprogrammeerd in de regels 900 t/m 990. De adressen 64 en 65 worden gebruikt, dus 191 en 190 in geïnverteerde vorm. Regel 910 is reeds besproken. De uitgang wordt transparant gemaakt, zodat ze veranderingen in het dataregister gaat volgen. Hierin worden steeds weer opnieuw achtereenvolgens alle waarden van de functie zoals opgeslagen in de tabel gebracht. Onderbreking is echter mogelijk door het indrukken van een willekeurige toets (regel 940) en dan wordt in regel 980 de uitgestuurde spanning nul gemaakt en vervolgens de DAC8 gesloten. Regel 950 is de aanroep van een vertragingsslus, waarvan de wachttijd afhankelijk is van de waarde van N, die bepaalt of de functie snel, matig of traag moet zijn.

Omdat het programma door de BASIC-interpretator wordt uitgevoerd is de snelheid niet hoog. Op de NMS8250 betekent snel, matig en traag periodes tijden van resp. 1,6, 3,4 en 5,6 seconden. Een snelle functiegenerator is te maken door de DAC8 aan te sturen in machinetaal. Ik heb daarmee al wat geëxperimenteerd op mijn NMS9100 en kwam tot 90000 uitgestuurde waarden per seconde. Voor het verkrijgen van een mooi signaal moet dan natuurlijk wel de interrupt afgezet worden, omdat anders 19 keer per seconde het schermonderhoud de functie verstoort.



```

5 REM *****
10 REM * Naam programma : FUNGEN.BAS *
15 REM * Aanmaakdatum : 9 JANUARI 1990 *
20 REM * Computer : MSX2 *
25 REM * Hardware : Komeet-systeem met DAC8 print. *
30 REM * Beschrijving : De DAC8 stuurt een spanningsfunctie uit met een *
35 REM * vorm naar keuze, een amplitude naar keuze en een *
40 REM * frekwentie naar keuze. Onderbreking is mogelijk *
45 REM * door het intoetsen van een willekeurige toets. *
50 REM * Opmerkingen : Geschreven in MSX-BASIC. *
55 REM *****
100 DIM FUN(100)
110 COLOR 4,14,14:WIDTH 80:CLS:LOCATE 20,8:PRINT"FUNCTIEGENERATOR"
115 N=2000:GOSUB 1000
120 CLS:LOCATE 36,5:PRINT "M E N U"
130 LOCATE 15,9:PRINT "KIES DE FUNCTIE-VORM:"
140 LOCATE 17,10:PRINT "1 = SINUS"
150 LOCATE 17,11:PRINT "2 = BLOK"
160 LOCATE 17,12:PRINT "3 = ZAAGTAND OPLOPEND"
170 LOCATE 17,13:PRINT "4 = ZAAGTAND AFLOPEND"
180 LOCATE 17,14:PRINT "5 = DRIEHOEK"
185 LOCATE 17,15:PRINT "6 = STOPPEN"
190 LOCATE 17,20:INPUT "WAT IS UW KEUS";K:IF K=6 THEN END
200 IF K<1 OR K>5 THEN 120
210 CLS:LOCATE 10,5:PRINT "KIES DE AMPLITUDE"
220 LOCATE 10,7:PRINT "1 = 2.5 VOLT"
230 LOCATE 10,8:PRINT "2 = 2.0 VOLT"
240 LOCATE 10,9:PRINT "3 = 1.5 VOLT"
250 LOCATE 10,10:PRINT "4 = 1.0 VOLT"
260 LOCATE 10,11:PRINT "5 = 0.5 VOLT"
270 LOCATE 17,20:INPUT "WAT IS UW KEUS";AM:IF AM<1 OR AM>5 THEN 210
300 CLS:LOCATE 10,5:PRINT "KIES DE FREKWENTIE"
310 LOCATE 10,7:PRINT "1 = SNEL"
320 LOCATE 10,8:PRINT "2 = MATIG"
330 LOCATE 10,9:PRINT "3 = TRAAG"
340 LOCATE 17,20:INPUT "WAT IS UW KEUS";FR:IF FR<1 OR FR>3 THEN 300
390 ON K GOTO 400,500,600,700,800
400 PI=3.1415927#:X=(250-(AM-1)*50)/2
410 FOR I=1 TO 100
420 FUN(I)=INT(SIN(2*PI*I/100)*X+X)
430 NEXT I
440 GOTO 900
500 X=250-(AM-1)*50
510 FOR I=1 TO 50:FUN(I)=0:NEXT
520 FOR I=51 TO 100:FUN(I)=X:NEXT
530 GOTO 900
600 X=250-(AM-1)*50
610 FOR I=0 TO 99:FUN(I+1)=INT(I/99*X):NEXT
620 GOTO 900
700 X=250-(AM-1)*50
710 FOR I=0 TO 99:FUN(100-I)=INT(I/99*X):NEXT
720 GOTO 900
800 X=250-(AM-1)*50
810 FOR I=0 TO 49:FUN(I+1)=INT(I/49*X):NEXT
820 FOR I=50 TO 99:FUN(I+1)=INT((100-I)/49*X):NEXT
830 GOTO 900
900 N=(FR-1)*10:CLS:LOCATE 20,8:PRINT"DE DAC8 STUURT SIGNAAL UIT"
910 OUT 49,191:OUT 48,0:OUT 49,190
920 FOR I=1 TO 100
930 OUT 48,FUN(I)
940 AS=INKEY$:IF AS<>" THEN 980
950 GOSUB 1000
960 NEXT I
970 GOTO 920
980 OUT 48,0:OUT 49,191:OUT 48,15:OUT 49,0
990 GOTO 110
1000 FOR J=0 TO N:NEXT J
1001 RETURN

```